NAG Library Routine Document

M01DBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

M01DBF ranks a vector of integer numbers in ascending or descending order.

2 Specification

```
SUBROUTINE M01DBF (IV, M1, M2, ORDER, IRANK, IFAIL)
INTEGER      IV(M2), M1, M2, IRANK(M2), IFAIL
CHARACTER(1) ORDER
```

3 Description

M01DBF uses a variant of list-merging, as described on pages 165–166 in Knuth (1973). The routine takes advantage of natural ordering in the data, and uses a simple list insertion in a preparatory pass to generate ordered lists of length at least 10. The ranking is stable: equal elements preserve their ordering in the input data.

4 References

Knuth D E (1973) *The Art of Computer Programming (Volume 3)* (2nd Edition) Addison–Wesley

5 Arguments

1: IV(M2) – INTEGER array *Input*

On entry: elements M1 to M2 of IV must contain integer values to be ranked.

2: M1 – INTEGER *Input*

On entry: the index of the first element of IV to be ranked.

Constraint: M1 > 0.

3: M2 – INTEGER *Input*

On entry: M2 must specify the index of the last element of IV to be ranked.

Constraint: M2 $\geq$ M1.

4: ORDER – CHARACTER(1) *Input*

On entry: if ORDER = 'A', the values will be ranked in ascending (i.e., nondecreasing) order.

If ORDER = 'D', into descending order.

Constraint: ORDER = 'A' or 'D'.

5: IRANK(M2) – INTEGER array *Output*

On exit: elements M1 to M2 of IRANK contain the ranks of the corresponding elements of IV. Note that the ranks are in the range M1 to M2: thus, if IV(i) is the first element in the rank order, IRANK(i) is set to M1.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, M2 $<$ 1,
or M1 $<$ 1,
or M1 $>$ M2.

IFAIL $= 2$

On entry, ORDER is not `A' or `D'.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

M01DBF is not threaded in any implementation.

9 Further Comments

The average time taken by the routine is approximately proportional to $n \times \log(n)$, where $n = \text{M2} - \text{M1} + 1$.

10 Example

This example reads a list of integers and ranks them in descending order.

10.1 Program Text

```
    Program m01dbfe

!     M01DBF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: m01dbf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, m1, m2
!     .. Local Arrays ..
      Integer, Allocatable             :: irank(:), iv(:)
!     .. Executable Statements ..
      Write (nout,*) 'M01DBF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) m2
      Allocate (irank(m2),iv(m2))

      m1 = 1

      Read (nin,*)(iv(i),i=m1,m2)

      ifail = 0
      Call m01dbf(iv,m1,m2,'Descending',irank,ifail)

      Write (nout,*)
      Write (nout,*) '   Data   Ranks'
      Write (nout,*)

      Do i = m1, m2
        Write (nout,99999) iv(i), irank(i)
      End Do

99999 Format (1X,2I7)
    End Program m01dbfe
```

10.2 Program Data

```
M01DBF Example Program Data
12
34 44 89 64 69 69 23 1 999 65 22 76
```

10.3 Program Results

```
 M01DBF Example Program Results

    Data   Ranks

      34      9
      44      8
      89      2
      64      7
      69      4
      69      5
      23     10
```

```
     1      12
   999       1
    65       6
    22      11
    76       3
```